
Subject: stacking the images according to julian date
Posted by [ansul](#) on Tue, 20 Aug 2013 20:10:27 GMT
[View Forum Message](#) <> [Reply to Message](#)

Hello all,

I am a newbie in IDL and thought I might get some information here.

I need to do a time series analysis with landsat images (almost 500 images). I need to stack the image according to the julian dates. So I have 500 images in a folder which are named Indsr.LT50140331989278XXX01. In this naming format the Julian date starts from the 20th position to 22nd position. And I need to use this information to stack the image.

All your help will be much appreciated.

Thanks

Subject: Re: stacking the images according to julian date
Posted by [David Fanning](#) on Tue, 20 Aug 2013 20:27:24 GMT
[View Forum Message](#) <> [Reply to Message](#)

ansul writes:

> I am a newbie in IDL and thought I might get some information here.
>
> I need to do a time series analysis with landsat images (almost 500 images). I need to stack the image according to the julian dates. So I have 500 images in a folder which are named Indsr.LT50140331989278XXX01. In this naming format the Julian date starts from the 20th position to 22nd position. And I need to use this information to stack the image.
>
> All your help will be much appreciated.

```
filenames = File_Search("Indsr.*", COUNT=count)  
juldates = DblArr(count)
```

```
; Parse the filename for the Julian date.  
FOR j=0,count-1 DO BEGIN  
  thisFile = filenames[j]  
  year = Fix(StrMid(thisFile, 15, 4))  
  doy = Fix(StrMid(thisFile, 19, 3))  
  juldates[j] = JulDay(1, doy, year)  
ENDFOR
```

```
; Sort the Julian dates.  
sorted = Sort(juldates)
```

```
; Sort the filenames in the right order for stacking.
```

```
filenames = filenames[sorted]
```

I don't think you REALLY want to stack these images, but now you have them in the right order, in case you want to process them in the right order.

Cheers,

David

--

David Fanning, Ph.D.

Fanning Software Consulting, Inc.

Coyote's Guide to IDL Programming: <http://www.idlcoyote.com/>

Sepore ma de ni thue. ("Perhaps thou speakest truth.")

Subject: Re: stacking the images according to julian date
Posted by [wlandsman](#) on Tue, 20 Aug 2013 21:02:34 GMT
[View Forum Message](#) <> [Reply to Message](#)

And to be an annoying kibbutzer-- I believe this can also be done without loops:

```
filenames = File_Search("Indsr.*", COUNT=count)  
year = Fix(StrMid(filenames,15,4))  
doy = Fix(StrMid(filenames,19,3))  
juldates = JulDay(1,doy, year)
```

```
; Sort the Julian dates.  
sorted = Sort(juldates)
```

```
; Sort the filenames in the right order for stacking.
```

```
filenames = filenames[sorted]
```

On Tuesday, August 20, 2013 4:27:24 PM UTC-4, David Fanning wrote:

```
>>  
>  
>> All your help will be much appreciated.  
>  
>  
>  
> filenames = File_Search("Indsr.*", COUNT=count)  
>  
> juldates = DbfArr(count)
```

```
>
>
>
> ; Parse the filename for the Julian date.
>
> FOR j=0,count-1 DO BEGIN
>
>   thisFile = filenames[j]
>
>   year = Fix(StrMid(thisFile, 15, 4))
>
>   doy = Fix(StrMid(thisFile, 19, 3))
>
>   juldates[j] = JulDay(1, doy, year)
>
> ENDFOR
>
>
>
> ; Sort the Julian dates.
>
> sorted = Sort(juldates)
>
>
>
> ; Sort the filenames in the right order for stacking.
>
> filenames = filenames[sorted]
>
>
>
> I don't think you REALLY want to stack these images, but now you have
>
> them in the right order, in case you want to process them in the right
>
> order.
>
>
>
> Cheers,
>
>
>
> David
>
>
>
> --
```

>
> David Fanning, Ph.D.
>
> Fanning Software Consulting, Inc.
>
> Coyote's Guide to IDL Programming: <http://www.idlcoyote.com/>
>
> Sepore ma de ni thue. ("Perhaps thou speakest truth.")

Subject: Re: stacking the images according to julian date
Posted by [David Fanning](#) on Tue, 20 Aug 2013 21:17:41 GMT
[View Forum Message](#) <> [Reply to Message](#)

Wayne Landsman writes:

> And to be an annoying kibbutzer-- I believe this can also be done without loops:
>
> filenames = File_Search("Insr.*", COUNT=count)
> year = Fix(StrMid(filenames,15,4))
> doy = Fix(StrMid(filenames,19,3))
> juldates = JulDay(1,doy, year)
>
> ; Sort the Julian dates.
> sorted = Sort(juldates)
>
> ; Sort the filenames in the right order for stacking.
>
> filenames = filenames[sorted]

Even better, if harder for a newby to understand! ;-)

Cheers,

David

P.S. It's even hard for old guys to understand. I almost always have to read your article on STRMID to understand how it works on arrays.

--

David Fanning, Ph.D.
Fanning Software Consulting, Inc.
Coyote's Guide to IDL Programming: <http://www.idlcoyote.com/>
Sepore ma de ni thue. ("Perhaps thou speakest truth.")
